

PROGRAMAÇÃO E ALGORITMOS (PEA)

AULA 04

Professor(es): ADRIANO DOIMO / CARLOS REBOLLO

E-MAIL: adriano.doimo@etec.sp.gov.br / carlos.rebollo@etec.sp.gov.br

AULA 04

- Técnica *Top-Down*
- Estruturas de Seleção
 - » Seleção Simples
 - » Seleção Composta
 - » Seleção Encadeada

ANÁLISE E RESOL. PROBLEMAS

- Para a criação de qualquer algoritmo, alguns passos básicos devem ser seguidos:
 - » Entender o problema
 - » Definir as variáveis
 - Variáveis de entrada
 - Variáveis de saída
 - » Definir o processamento
 - » Escrever o algoritmo

ANÁLISE E RESOLUÇÃO DE PROBLEMAS


- Problemas comuns na análise e resolução de problemas.
 - » Ler um pedaço do enunciado e achar que já entendeu tudo.
 - » Ler um pedaço do enunciado e achar que não vai entender nada.
- Qual dos dois é pior?
 - » OS DOIS!!!

TÉCNICAS TOP-DOWN

- É uma técnica natural (intuitivamente utilizada por nós) para a resolução de problemas complexos:
- Consiste em “quebrar” um problema mais complexo e deixa-los “menores” e mais fáceis de gerenciar, repetindo o processo até chegarmos a um tamanho de problema com o qual sabemos lidar.

ESTRUTURAS DE SELEÇÃO

- Permitem que uma escolha seja feita durante a execução de um algoritmo, de modo que um determinado trecho de código seja ou não executado de acordo com o resultado de uma expressão lógica.




ESTRUTURAS DE SELEÇÃO

- Formato - Seleção Simples

```
se <condição> entao  
    <bloco de comandos>  
fimse
```

Descrição:

Executa o <bloco de comandos> apenas **se** a expressão lógica <condição> retornar VERDADEIRO.



ESTRUTURAS DE SELEÇÃO

- Seleção Simples

```
algoritmo "selecao_simples"  
var  
    N1, N2, N3, N4, MA : real  
inicio  
    escreva ("Informe as 4 notas: ")  
    leia (N1, N2, N3, N4)  
    MA <- (N1 + N2 + N3 + N4) / 4  
    escreval ("Media Anual= ", MA)  
    se (MA >= 5) entao  
        escreval ("Aluno Aprovado!")  
    fimse  
fimalgoritmo
```

```
Informe as 4 notas:  
5  
6  
7  
8  
Media Anual= 6.5  
Aluno Aprovado!  
  
*** Fim da execução.  
*** Feche esta janela para retornar ao Visualg.
```



ESTRUTURAS DE SELEÇÃO

- Formato - Seleção Composta

```
se <condição> entao  
    <bloco de comandos1>  
senao  
    <bloco de comandos2>  
fimse
```

Descrição:

Executa o <bloco de comandos1> **se** a expressão lógica <condição> retornar VERDADEIRO **senao** o <bloco de comandos2> se retornar FALSO.



ESTRUTURAS DE SELEÇÃO

- Seleção Composta

algoritmo "selecao_composta"

var

N1, N2, N3, N4, MA : real

inicio

escreval ("Informe as 4 notas: ")

leia (N1, N2, N3, N4)

MA <- (N1 + N2 + N3 + N4) / 4

escreval ("Media Anual= ", MA)

se (MA >= 5) entao

 escreval ("Aluno Aprovado!")

 escreval ("Parabéns!")

senao

 escreval ("Aluno Reprovado")

 escreval ("Que pena...")

fimse

fimalgoritmo

Informe as 4 notas:

1

1

1

1

Media Anual= 1

Aluno Reprovado

Que pena...

*** Fim da execução.

*** Feche esta janela para retornar ao Visualg.

ESTRUTURAS DE SELEÇÃO

- Seleção Encadeada

Descrição:

As seleções podem se tornar tão complexas quanto forem necessárias.

```
se <condição1> entao
    se <condição 2> entao
        <bloco de comandos1>
    senao
        <bloco de comandos2>
fimse
fimse
```

```
se <condição1> entao
    <bloco de comandos1>
senao
    se <condição2> entao
        <bloco de comandos2>
    senao
        <bloco de comandos3>
fimse
fimse
```

PRATICANDO...

- EXERCÍCIO.
- Elabore um algoritmo que solicite um valor entre 10 e 20 e informe se:
 - » O valor está correto
 - » O valor é menor que 10
 - » O valor é maior que 20.

PRATICANDO...

- RESPOSTA.

```
algoritmo "selecao_composta"  
var  
    num: inteiro  
inicio  
    escreval ("Digite uma valor entre 10 e 20 ")  
    leia(num)  
    se ((num>10) e (num<20)) entao  
        escreval ("O valor está correto")  
    senao  
        se (num<10) entao  
            escreval ("O valor é menor que 10")  
        senao  
            escreval ("O valor é maior que 20")  
        fimse  
    fimse  
fimalgoritmo
```

```
Digite uma valor entre 10 e 20  
15  
O valor está correto  
  
*** Fim da execução.  
*** Feche esta janela para reto
```

```
Digite uma valor entre 10 e 20  
5  
O valor é menor que 10  
  
*** Fim da execução.  
*** Feche esta janela para reto
```

```
Digite uma valor entre 10 e 20  
25  
O valor é maior que 20  
  
*** Fim da execução.  
*** Feche esta janela para reto
```


ESTRUTURAS DE SELEÇÃO

O que é um triângulo?

R: Figura geométrica de três lados, em que cada lado é menor do que a soma dos outros dois lados.


O que é um triângulo equilátero?

R: Um triângulo com três lados iguais.




O que é um triângulo isósceles?

R: Um triângulo com dois lados iguais.



O que é um triângulo escaleno?

R: Um triângulo com todos os lados diferentes.



ESTRUTURAS DE SELEÇÃO

- Dados três valores A , B , C , verificar se eles podem ser os comprimentos dos lados de um triângulo e, se forem, verificar se compõem um triângulo equilátero, isósceles ou escaleno.
- Informar também, se estes valores não compuserem nenhum triângulo.
 - » Dados de entrada: três lados de um suposto triângulo (A , B , C).
 - » Dados de saída: mensagens: não compõem triângulo, triângulo equilátero, triângulo isósceles, triângulo escaleno.

ESTRUTURAS DE SELEÇÃO

- Traduzindo as condições para a tabela-verdade...

É triângulo?	É equilátero?	É isósceles?	É escaleno?	Ações
V	V	F	F	“Equilátero”
V	F	V	-	“Isósceles”
V	F	F	V	“Escaleno”
F	-	-	-	“Não é triângulo”

ESTRUTURAS DE SELEÇÃO

- Traduzindo as condições para expressões lógicas...
 - » É triângulo: $(A < B + C) \text{ e } (B < A + C) \text{ e } (C < A + B)$
 - » É equilátero: $(A = B) \text{ e } (B = C)$
 - » É isósceles: $(A = B) \text{ ou } (A = C) \text{ ou } (B = C)$
 - » É escaleno: $(A <> B) \text{ e } (B <> C) \text{ e } (A <> C)$

RESPOSTA...

```
algoritmo "triangulo"  
var  
A, B, C: inteiro  
inicio  
escreval ("Informe os lados do triangulo: ")  
leia(A, B, C)  
se ((A < B + C) e (B < A + C) e (C < A + B)) entao  
    escreval ("Estes valores formam um triangulo!")  
    se ((A = B) e (B = C)) entao  
        escreval ("Triangulo Equilatero")  
    senao  
        se ((A = B) ou (A = C) ou (B = C)) entao  
            escreval ("Triangulo Isosceles")  
        senao  
            escreval ("Triangulo Escaleno")  
        fimse  
    fimse  
senao  
    escreval ("Estes valores não formam um triangulo!")  
fimse  
fimalgoritmo
```

ALGUNS RESULTADOS...

```
Informe os lados do triangulo:
5
5
5
Estes valores formam um triangulo!
Triangulo Equilatero

*** Fim da execução.
*** Feche esta janela para retornar
```

```
Informe os lados do triangulo:
5
4
3
Estes valores formam um triangulo!
Triangulo Escaleno

*** Fim da execução.
*** Feche esta janela para retornar
```

```
Informe os lados do triangulo:
5
5
4
Estes valores formam um triangulo!
Triangulo Isosceles

*** Fim da execução.
*** Feche esta janela para retornar
```

```
Informe os lados do triangulo:
10
5
4
Estes valores não formam um triangulo!

*** Fim da execução.
*** Feche esta janela para retornar ao
```